3次元コンピューターグラフィックスと正多面体

磯　川　幸　直

(2003年10月21日　受理)

3-dimensional computer graphicd and regular polyhedra

Yukinao ISOKAWA

要約

数学教育における3次元コンピューターグラフィックスの利用法について，主にその技術的側面を考察する．具体的には，正多面体を題材にとり，3DCG ソフトウェア POV-Ray のためのコードを書く方法を示す．

1　はじめに

数学教育の現場において，3次元コンピューターグラフィックス (3DCG) は，毎日の授業の質を高めるのに有用な道具を提供する．しかし，これを用いるための必要な技術は，ふつうの教員養成課程のカリキュラムの枠外にあるため，教師が 3DCG 使ってみたいと思っても，諦めざるをえないのが現状である．この論説では，数学教育において 3DCG をどのように用いることができるか，を示す1つの例を提供したい．

3DCG を適用する題材には，正多面体を選ぶことにした．と言うのは，これが視覚的に美しく，3DCG に適していることにもよるが，それ以上に正多面体が，中学校で扱われる程度に初等的であるにもかかわらず，豊富な数学的内容をもっているからである．

ところで，正多面体を 3DCG の題材にしたホームページは数多く見つけることができる．ところがこれらのホームページでは，正多面体の画像が示されているだけで，それら画像の作成方法は説明されていない．とくに筆者が調べた範囲では，この論説で扱う 3DCG ソフトウェア POV-Ray のコードが具体的に示されているホームページを見つけることはできなかった．

この記事の目的は，3DCG を用いた数学教育の方法を例示することであるが，学習のすべてをコンピューター上で行うべきであると主張しているわけではない．その反対に筆者は，数学のどんな学習においても，精神と手と眼を並行して用いるべきであると考える．とくに正多面体を学習するためには，双対のような変換・変形を精神を集中させて施し，手で実際に模型を作り，その模型を注意深く観察する過程が不可欠である．ではコンピューターは学習においてどんな役割り果たすことができるか? 筆者は，次の3つの役割りを果たすと考える．

- コンピューターが理解できるコードには，曖昧さは許されない．そこで具体的にコードを書きくだすことにより，自分が持っている知識をデバッグすることができる．

- 実際に模型を作る作業では，その過程の途中で修正をすることは難しい．すなわち，作業を始める前にあらかじめ正確な設計図が必要であり，模型製作は対話性に乏しい．

- とくに3DCGでは，たとえ手先が不器用な人でも，美しい画像作ることができる．すなわち人間のもつ審美的欲求を比較的容易に満たしてくれる．

この論説においてはPOV-Rayというオープンソースのフリーソフトウェアを用いることにした．その理由は，このソフトウェアが著しい性能を持っているだけでなく，様々なOSの下で動くことにある．教育の現場においては，特定のOSの下でのみ動くソフトウェアを用いることは，学生の将来の選択肢を狭めることになり，不適切であろう．さらに大学において，教員または学生が，希望するならばいつでもソースコードを研究できる環境が重要である．オープンソースでないソフトウェアを用いることは，研究し学習する場としての大学を崩壊させることになる．

2　立方体，正8面体，そして正4面体

2.1　立方体

立方体の頂点の座標は，たとえば $(\pm1,\pm1,\pm1)$ で与えることができる．POV-Rayでは多面体を，その頂点の座標と 'triangle' プリミティブを用いて，定義することができる(POV-Rayのキーワードについては附録Aを参照)．しかし，POV-RayのもつCSG (構成的立体幾何) の機能を十分に活用したいならば，, 多面体を 'plane' プリミティブを用いて，半空間の共通部分として定義しなければいけない．そこで立方体を次のように定義する：

```
#declare Cube = intersection
{
   plane { < 1, 0, 0>, 1 }
   plane { <-1, 0, 0>, 1 }
   plane { <0,  1, 0>, 1 }
   plane { <0, -1, 0>, 1 }
   plane { <0, 0,  1>, 1 }
   plane { <0, 0, -1>, 1 }
}
```

これはソリッド(中身の詰まった)な立方体である．ときには，多面体を針金(もしくは竹ひご)で作ることが必要になる．そのためには 'cylinder' プリミティブを用いるとよい．

```
#declare b = 0.02;  // cylinder の半径
#declare WiredCube = union
{
   cylinder { <1,1,1>, <-1,1,1>, b }
   cylinder { <1,1,1>, <1,-1,1>, b }
   cylinder { <1,1,1>, <1,1,-1>, b }
   cylinder { <-1,1,1>, <-1,-1,1>, b }
   cylinder { <-1,1,1>, <-1,1,-1>, b }
   cylinder { <1,-1,1>, <-1,-1,1>, b }
```

```
    cylinder { <1,-1,1>, <1,-1,-1>, b }
    cylinder { <1,1,-1>, <-1,1,-1>, b }
    cylinder { <1,1,-1>, <1,-1,-1>, b }
    cylinder { <-1,-1,-1>, <1,-1,-1>, b }
    cylinder { <-1,-1,-1>, <-1,1,-1>, b }
    cylinder { <-1,-1,-1>, <-1,-1,1>, b }
}
```

さて，任意の正多面体に対して，３つの重要な球を考えることができる．

1. 多面体のすべての頂点を通る**外接球.**
2. 多面体のすべての辺に接する**中接球.**
3. 多面体のすべての面に接する**内接球.**

立方体に対しては，

$$\text{内接球の半径} : \text{中接球の半径} : \text{外接球の半径} = 1 : \sqrt{2} : \sqrt{3}$$

となる．

2.2 立方体の内部にある２つの正４面体

立方体の８個の頂点を，次の２つのグループに分ける：

$$(1,1,1),(1,-1,-1),(-1,1,-1),(-1,-1,1)$$

と

$$(-1,-1,-1),(-1,1,1),(1,-1,1),(1,1,-1)\ .$$

すると，この２つのグループのそれぞれが正４面体を作る．
そこで，第１の正４面体はつぎのように定義できる：

```
#declare k = 0.57735072; // = 1 / sqrt(3)
#declare Tetrahedron = intersection
{
   plane { <1, 1, -1>, k }
   plane { <1, -1, 1>, k }
   plane { <-1, 1, 1>, k }
   plane { <-1, -1, -1>, k }
}
```

他方，第２の正４面体は，第１の正４面体を原点に関して点対称に移動したものである．そこで，次のように定義できる：

```
object { Tetrahedron  scale <-1, -1, -1> }
```

次の図は，立方体の内部にある２つの正４面体を描いている．



2.3 正８面体

２個の正４面体の共通部分は，ソリッドな正８面体になる．

```
#declare Octahedron = intersection
{
   object { Tetrahedron }
   object { Tetrahedron  scale <-1,-1,-1> }
}
```

ところで，６個の点

$$(\pm1,0,0),(0,\pm1,0),(0,0,\pm1),$$

を適切に辺で結ぶと，上で作ったソリッドな正８面体とピッタリ重なる，針金で作った正８面体ができる．

```
#declare WiredOctahedron = union
{
   cylinder { <1,0,0>, <0,1,0>, b }
   cylinder { <1,0,0>, <0,0,1>, b }
   cylinder { <1,0,0>, <0,-1,0>, b }
   cylinder { <1,0,0>, <0,0,-1>, b }
```

```
    cylinder { <0,1,0>, <0,0,1>, b }
    cylinder { <0,0,1>, <0,-1,0>, b }
    cylinder { <0,-1,0>, <0,0,-1>, b }
    cylinder { <0,0,-1>, <0,1,0>, b }
    cylinder { <-1,0,0>, <0,1,0>, b }
    cylinder { <-1,0,0>, <0,0,1>, b }
    cylinder { <-1,0,0>, <0,-1,0>, b }
    cylinder { <-1,0,0>, <0,0,-1>, b }
}
```

正8面体に対しては，

$$\text{内接球の半径} : \text{中接球の半径} : \text{外接球の半径} = \frac{1}{\sqrt{3}} : \frac{1}{\sqrt{2}} : 1$$

となる．

そこで，同じ中接球をもつ針金の立方体とソリッドな正8面体を同時に描きたいならば，ソリッドな正8面体を原点に関して2倍に相似拡大すればよい．こうして描かれた次の図は，この2つの立体がたがいに双対であることを鮮かに示している．



ここで，CSG のパワーを示すいくつかの問題を提出しよう．問題の解答は附録Bで与える．

問題1　同じ中接球をもつソリッドな立方体とソリッドな正8面体の共通部分を描きなさい．それはどんな多面体かな?

問題2　上の2個の正4面体の和を描きなさい．それはどんな多面体かな?

2.4 双対

正多面体 $\{p,q\}$ を考え，それが V 個の頂点，E 本の辺，F 枚の面からできているとしよう．その多面体の各辺を，それを垂直に2等分しかつ中接球と接する辺で置き換えるならば，F 個の頂点，E 本の辺，V 枚の面からできている双対多面体 $\{q,p\}$ (正確には中接球に関する双対多面体) を得る．

多面体 $\{p,q\}$ の内接球，中接球，外接球の半径がそれぞれ r,ρ,R であり，またそれに双対な多面体 $\{q,p\}$ の内接球，中接球，外接球の半径がそれぞれ r',ρ',R' であるならば，次の関係が成立する．

$$rR' = Rr' = 1. \tag{1}$$

3 正20面体と正12面体

3.1 正20面体

次節で説明するように，正20面体の12個の頂点の座標はつぎのように与えることができる：

$$(\pm\tau,\pm1,0),(0,\pm\tau,\pm1),(\pm1,0,\pm\tau) \tag{2}$$

このとき20枚の面の法線ベクトルは

$$(\pm1,\pm1,\pm1),(\pm\tau,0,\pm1/\tau),(\pm1/\tau,\pm\tau,0),(0,\pm1/\tau,\pm\tau) \tag{3}$$

となる．上記のデータをもとに計算すると，正20面体に対して，

$$\text{内接球の半径}:\text{中接球の半径}:\text{外接球の半径} = \frac{\tau}{\sqrt{3}}:1:\frac{\sqrt{\tau^2+1}}{\tau}. \tag{4}$$

となることがわかる．

面の法線ベクトルのデータ (3) より，内接球の半径が1であるソリッドな正20面体はつぎのように定義できる：

```
#declare tau = 1.618033989;
#declare inv = 0.6180339887; // = 1 / tau

#declare p1  = <1, 1, 1>;
#declare p2  = <1, 1, -1>;
#declare p3  = <1, -1, 1>;
#declare p4  = <1, -1, -1>;
#declare p5  = <-1, 1, 1>;
#declare p6  = <-1, 1, -1>;
#declare p7  = <-1, -1, 1>;
```

```
#declare p8  = <-1, -1, -1>;
#declare p9  = <tau, 0, inv>;
#declare p10 = <tau, 0, -inv>;
#declare p11 = <-tau, 0, inv>;
#declare p12 = <-tau, 0, -inv>;
#declare p13 = <inv, tau, 0>;
#declare p14 = <-inv, tau,0>;
#declare p15 = <inv, -tau, 0>;
#declare p16 = <-inv, -tau, 0>;
#declare p17 = <0, inv, tau>;
#declare p18 = <0, -inv, tau>;
#declare p19 = <0, inv, -tau>;
#declare p20 = <0, -inv, -tau>;

#declare Icosahedron = intersection
{
   plane { p1, 1 }
   plane { p2, 1 }
   plane { p3, 1 }
   plane { p4, 1 }
   plane { p5, 1 }
   plane { p6, 1 }
   plane { p7, 1 }
   plane { p8, 1 }
   plane { p9, 1 }
   plane { p10, 1 }
   plane { p11, 1 }
   plane { p12, 1 }
   plane { p13, 1 }
   plane { p14, 1 }
   plane { p15, 1 }
   plane { p16, 1 }
   plane { p17, 1 }
   plane { p18, 1 }
   plane { p19, 1 }
   plane { p20, 1 }
}
```

一方，頂点の座標のデータ (2) より，針金で作った正 20 面体は次のようになる：

```
#declare q1  = <tau, 1, 0>;
```

```
#declare q2  = <tau, -1, 0>;
#declare q3  = <-tau, 1, 0>;
#declare q4  = <-tau, -1, 0>;
#declare q5  = <0, tau, 1>;
#declare q6  = <0, tau, -1>;
#declare q7  = <0, -tau, 1>;
#declare q8  = <0, -tau, -1>;
#declare q9 = <1, 0, tau>;
#declare q10 = <-1, 0, tau>;
#declare q11 = <1, 0, -tau>;
#declare q12 = <-1, 0, -tau>;

#declare WiredIcosahedron = union
{
   cylinder { q1,  q2,  b }
   cylinder { q1,  q5,  b }
   cylinder { q1,  q6,  b }
   cylinder { q1,  q9,  b }
   cylinder { q1,  q11, b }
   cylinder { q2,  q7,  b }
   cylinder { q2,  q8,  b }
   cylinder { q2,  q9,  b }
   cylinder { q2,  q11, b }
   cylinder { q3,  q4,  b }
   cylinder { q3,  q5,  b }
   cylinder { q3,  q6,  b }
   cylinder { q3,  q10, b }
   cylinder { q3,  q12, b }
   cylinder { q4,  q7,  b }
   cylinder { q4,  q8,  b }
   cylinder { q4,  q10, b }
   cylinder { q4,  q12, b }
   cylinder { q5,  q6,  b }
   cylinder { q5,  q9,  b }
   cylinder { q5,  q10, b }
   cylinder { q6,  q11, b }
   cylinder { q6,  q12, b }
   cylinder { q7,  q8,  b }
   cylinder { q7,  q9,  b }
```

```
    cylinder { q7,  q10, b }
    cylinder { q8,  q11, b }
    cylinder { q8,  q12, b }
    cylinder { q9,  q10, b }
    cylinder { q11, q12, b }
}
```

3.2 正12面体

正20面体の双対を作れば，正12面体が得られる．そこで針金で作った正20面体の定義から，ソリッドな正12面体の定義が導かれる：

```
#declare Dodecahedron = intersection
{
    plane { q1,  1 }
    plane { q2,  1 }
    plane { q3,  1 }
    plane { q4,  1 }
    plane { q5,  1 }
    plane { q6,  1 }
    plane { q7,  1 }
    plane { q8,  1 }
    plane { q9,  1 }
    plane { q10, 1 }
    plane { q11, 1 }
    plane { q12, 1 }
}
```

ソリッドな正20面体の定義からは，針金で作った正12面体の定義が導かれる：

```
#declare WiredDodecahedron = union
{
    cylinder { p1,  p9,  b }
    cylinder { p1,  p13, b }
    cylinder { p1,  p17, b }
    cylinder { p2,  p10, b }
    cylinder { p2,  p13, b }
    cylinder { p2,  p19, b }
    cylinder { p3,  p9,  b }
    cylinder { p3,  p15, b }
    cylinder { p3,  p18, b }
```

```
    cylinder { p4,  p10, b }
    cylinder { p4,  p15, b }
    cylinder { p4,  p20, b }
    cylinder { p5,  p11, b }
    cylinder { p5,  p14, b }
    cylinder { p5,  p17, b }
    cylinder { p6,  p12, b }
    cylinder { p6,  p14, b }
    cylinder { p6,  p19, b }
    cylinder { p7,  p11, b }
    cylinder { p7,  p16, b }
    cylinder { p7,  p18, b }
    cylinder { p8,  p12, b }
    cylinder { p8,  p16, b }
    cylinder { p8,  p20, b }
    cylinder { p9,  p10, b }
    cylinder { p11, p12, b }
    cylinder { p13, p14, b }
    cylinder { p15, p16, b }
    cylinder { p17, p18, b }
    cylinder { p19, p20, b }
}
```

(1) と (4) より，正12面体に対しては，

$$\text{内接球の半径 : 中接球の半径 : 外接球の半径} = \frac{\tau}{\sqrt{\tau^2+1}} : 1 : \frac{\sqrt{3}}{\tau}. \tag{5}$$

となる．

さて，同じ中接球半径をもつ針金で作った正20面体とソリッドな正12面体を同時に描くことにしよう．WiredIcosahedron の定義より，その外接球の半径は $\sqrt{\tau^2+1}$，したがって (4) から，中接球の半径は τ である．一方，Dodecahedron の定義より，その内接球の半径は 1，したがって (5) から，中接球の半径は $\sqrt{\tau^2+1}/\tau$ である．したがって，中接球半径を等しくするには，正12面体を $\frac{\tau^2}{\sqrt{\tau^2+1}}$ 倍だけ相似拡大すればよい．こうして描かれた図は，この２つの多面体がたがいに双対であることを示している．



問題3　同じ中接球半径をもつ針金で作った正 12 面体とソリッドな正 20 面体を同時に描きなさい．

問題4　同じ中接球半径をもつソリッドな正 12 面体とソリッドな正 20 面体の共通部分を描きなさい．これはどんな多面体かな?

4　5つの正多面体の間の諸関係

4.1　正8面体に内接する正20面体

正8面体が，2個の正4面体から作られた事を思い出そう．1つの正4面体の面を白く塗り，もう1つの1つの正4面体の面を黒く塗ると，正8面体の隣り合う面は個となる色で塗られる．白い面に時計回りの輪を描き，黒い面に反時計回りの輪を描く．すると，正8面体の各辺に，矛盾無く向きを付けることができる．もしあなたがその向きに沿って辺の上をを動くならば，右側に白い面が，左側に黒い面が見えるであろう．

この辺の向き付けにしたがって，各辺を $1-t:t$ に内分した点を考えよう (ただし $0<t<1$)．この 12 個の内分点から，正8面体の各面の上で1個ずつ正3角形ができ，また正8面体の各頂点の場所で2個ずつ2等辺3角形ができる．こうして非正 20 面体ができる．

正8面体の頂点の座標が

$$(\pm1,0,0),(0,\pm1,0),(0,0,\pm1),$$

であるならば，この非正20面体の頂点の座標は

$$(\pm(1-t), \pm t, 0), (0, \pm(1-t), \pm t), (\pm t, 0, \pm(1-t)). \tag{6}$$

となる．非正20面体の8枚の面は，正8面体の面と一致するので，その方程式は

$$\pm x \pm y \pm z = 1.$$

である．他方，非正20面体の12枚の面の方程式は

$$\pm k_1 x \pm k_2 z = h,\ \pm k_1 y \pm k_2 x = h,\ \pm k_1 z \pm k_2 y = h \tag{7}$$

となる．ただし $k_1 = 1-t,\ k_2 = 1-2t, h = (1-t)^2$ である．

以上のデータより，POV-Ray の ‘clock’ 変数を用いて，非正20面体の形成をアニメーションにすることができる．具体的には，次のようにすればよい．

```
#declare t1 = 0.5 * clock;
#declare k1 = 1 - t1;
#declare k2 = 1 - 2 * t1;
#declare h = (1 - t1) * (1 - t1) / sqrt(k1 * k1 + k2 * k2);
#declare k = 0.57735072; // = 1 / sqrt(3)

#declare IrregularIcosahedron = intersection
{
   plane { <1, 1, -1>, k }
   plane { <1, -1, 1>, k }
   plane { <-1, 1, 1>, k }
   plane { <-1, -1, -1>, k }
   plane { <1, 1, 1>, k }
   plane { <1, -1, -1>, k }
   plane { <-1, 1, -1>, k }
   plane { <-1, -1, 1>, k }

   plane { <k1, 0, k2>, h }
   plane { <k1, 0, -k2>, h }
   plane { <-k1, 0, k2>, h }
   plane { <-k1, 0, -k2>, h }

   plane { <k2, k1, 0>, h }
   plane { <-k2, k1,0>, h }
   plane { <k2, -k1, 0>, h }
   plane { <-k2, -k1, 0>, h }
```

```
    plane { <0, k2, k1>, h }
    plane { <0, -k2, k1>, h }
    plane { <0, k2, -k1>, h }
    plane { <0, -k2, -k1>, h }
}
```

$t=1/(\tau+1)$ であるとき, 正8面体の頂点に位置する2等辺3角形は正3角形となり，正20面体が作られる．そこで (6) と (7) で $t=1/(\tau+1)$ を代入すると，(2) と (3) が得られる.

t が $\frac{1}{2}$ に近づくとき，正8面体の頂点に位置する2等辺3角形は同一平面上に乗り，1つの正方形を作る．こうして立方8面体が作られる．

4.2 正12面体に内接する立方体

正8面体に内接する正20面体に双対を施すことにより，正12面体に内接する立方体を得ることができる．



このとき次の疑問が生じる：何個の立方体が正12面体に内接するであろうか？

正12面体は20個の頂点をもつ．今，1つの頂点に着目する．上の図に見られるように，その頂点から出る立方体の1辺は，正12面体の1つの面である正5角形の対角線になっている．ところ

が同じ面には，やはりその頂点から出る対角線はもう1本ある．そこでこの頂点を含む立方体はもう1個あることに気が付くであろう．

x 個の立方体が正12面体に内接しているとしてみよう．ここで数え上げの定石であるが，横に x 個，縦に20個の格子点の集まりを考えてみる．格子点は全部で $x \times 20$ 個ある．それらのうち，立方体に含まれる頂点の組合せに対応する格子点を赤く塗る．すると，1つの立方体は8個の頂点を含むから，赤い格子点の個数は $x \times 8$ 個である．一方，1個の頂点は2個の立方体に含まれるから，赤い格子点の個数は 20×2 個でもある．したがって，$8x = 40$，すなわち $x = 5$ であることがわかる．

実際，模型を作って見れば，次の5個の立方体があることがわかる．

立方体1	頂点 = $\{1,2,3,4,5,6,7,8\}$ 辺 = $\{\{1,2\},\{1,3\},\{1,5\},\{2,4\},\{2,6\},\{3,4\},$ $\{3,7\},\{4,8\},\{5,6\},\{5,7\},\{6,8\},\{7,8\}\}$ 面 = $\{\{1,2,3,4\},\{1,2,5,6\},\{1,3,5,7\},$ $\{2,4,6,8\},\{3,4,7,8\},\{5,6,7,8\}\}$
立方体2	頂点 = $\{2,7,9,12,14,15,17,20\}$ 辺 = $\{\{2,9\},\{2,14\},\{2,20\},\{7,12\},\{7,15\},\{7,17\},$ $\{9,15\},\{9,17\},\{12,14\},\{12,20\},\{14,17\},\{15,20\}\}$ 面 = $\{\{2,9,14,17\},\{2,9,15,20\},\{2,12,14,20\},$ $\{7,9,15,17\},\{7,12,14,17\},\{7,12,15,20\}\}$
立方体3	頂点 = $\{3,6,10,11,13,16,17,20\}$ 辺 = $\{\{3,10\},\{3,16\},\{3,17\},\{6,11\},\{6,13\},\{6,20\},$ $\{10,13\},\{10,20\},\{11,16\},\{11,17\},\{13,17\},\{16,20\}\}$ 面 = $\{\{3,10,13,17\},\{3,10,16,20\},\{3,11,16,17\},$ $\{6,10,13,20\},\{6,11,13,17\},\{6,11,16,20\}\}$
立方体4	頂点 = $\{4,5,9,12,13,16,18,19\}$ 辺 = $\{\{4,9\},\{4,16\},\{4,19\},\{5,12\},\{5,13\},\{5,18\},$ $\{9,13\},\{9,18\},\{12,16\},\{12,19\},\{13,19\},\{16,18\}\}$ 面 = $\{\{4,9,13,19\},\{4,9,16,18\},\{4,12,16,19\},$ $\{5,9,13,18\},\{5,12,13,19\},\{5,12,16,18\}\}$
立方体5	頂点 = $\{1,8,10,11,14,15,18,19\}$ 辺 = $\{\{1,10\},\{1,14\},\{1,18\},\{8,11\},\{8,15\},\{8,19\},$ $\{10,15\},\{10,19\},\{11,14\},\{11,18\},\{14,19\},\{15,18\}\}$ 面 = $\{\{1,10,14,19\},\{1,10,15,18\},\{1,11,14,18\},$ $\{8,10,15,19\},\{8,11,14,19\},\{8,11,15,18\}\}$

ただし数字はIcosahedronの定義の中の点pに対する添字を表す．

これらのデータを用いて，5個の立方体は附録Cのように定義できる．

問題5　5個の立方体の和を描きなさい．

問題6　5個の立方体の共通部分を描きなさい．

4.3　正12面体に内接する正4面体

1つの立方体には2個の正4面体が内接する．そこで1つの正12面体には，10個の正4面体が内接することになる．具体的には，

- 立方体1の内部には，2.2節で定義した Tetrahedron と，それを原点に関して点対称移動した正4面体がある．
- 立方体2の内部には，頂点 $\{2, 12, 15, 17\}$ をもつ正4面体 Tetrahedron2 と，それを原点に関して点対称移動した正4面体がある．
- 立方体3の内部には，頂点 $\{3, 11, 13, 20\}$ をもつ正4面体 Tetrahedron3 と，それを原点に関して点対称移動した正4面体がある．
- 立方体4の内部には，頂点 $\{4, 12, 13, 18\}$ をもつ正4面体 Tetrahedron4 と，それを原点に関して点対称移動した正4面体がある．
- 立方体5の内部には，頂点 $\{1, 11, 15, 19\}$ をもつ正4面体 Tetrahedron5 と，それを原点に関して点対称移動した正4面体がある．

このことから，10個の正4面体は附録Dのように定義できる．

10個の正4面体の共通部分は正20面体になる．

問題7　10個の正4面体の和を描きなさい．

4.4　正20面体に外接する正8面体

正12面体に内接する5個の立方体の双対を考えれば，正20面体に外接する5個の正8面体が得られる．その定義は容易で附録Eに述べる．

明らかに，5個の正8面体の共通部分は，再び正20面体になる．

問題8　5個の正8面体の和を描きなさい．

正８面体は２個の正４面体の共通部分であるから，正20面体には10個の正４面体が外接する．明らかに，10個の正４面体の共通部分は，正20面体である．

問題９　10個の正４面体の和をどんな多面体かな?

参考文献

[1] Coxeter,H.S.M. (1973), *Reguar Polytopes*, Dover.

[2] Cromwell,P.R. (1997), *Polyhedra*, Cambridge University Press.

[3] http://www.povray.org/documentation

附録A

この記事を読むのみ必要な最小限の POV-Ray のキーワードの説明を述べる．

triangle: ３角形は，３個の頂点の座標から定義できる．

$$triangle\ \{< x_1, y_1, z_1 >, < x_2, y_2, z_2 >, < x_3, y_3, z_3 >\}$$

CSG: CSG (構成的立体幾何) とは，より単純な立体たちに，'union', 'intersection', 'difference' という演算を施して，より複雑な立体を構成するシステムを意味する．

plane: 平面は，法線ベクトル $< x, y, z >$ と，原点から平面までの距離 h 用いて定義される．法線ベクトルは単位ベクトルである必要はない．

$$plane\ \{< x, y, z >, h\}$$

cylinder: 円柱は，その２つの底面の中心の座標 $< x_1, y_1, z_1 >, < x_2, y_2, z_2 >$ と，底面の半径 b を用いて定義できる．

$$cylinder\ \{< x_1, y_1, z_1 >, < x_2, y_2, z_2 >, b\}$$

clock: 'clock' 変数は，アニメーションの繰り返しを実現する．デフォールトでは 0 から 1 まで変化する．

scale: 原点に関して，x 軸，y 軸，z 軸方向にそれぞれ λ, μ, ν 倍に相似拡大する場合，次のように書く．

$$scale < \lambda, \mu, \nu >$$

附録B

問題1の解答　得られる立体は，8個の正3角形と6個の正方形の面をもつ多面体である．この多面体は，**立方8面体** (cuboctahedron) と呼ばれ，13種類のアルキメデス多面体のうちの1つである．

```
#declare Cuboctahedron = intersection
{
   object { Cube }
   object { Octahedron  scale <2, 2, 2> }
}
```



問題2の解答　**Kepler の8角星** (stella octangula) と呼ばれる，星形多面体が作られる．

```
#declare StellaOctangula = union
{
   object { Tetrahedron }
   object { Tetrahedron  scale <-1, -1, -1> }
}
```



問題3の解答 ソリッドな正20面体の内接球の半径は1だから，中接球の半径は $\frac{\sqrt{3}}{\tau}$．一方，針金で作った正12面体の外接球の半径は $\sqrt{3}$ であるから，中接球の半径は τ．したがって中接球の半径を等しくするためには，正20面体を $\frac{\tau^2}{\sqrt{3}}$ 倍に相似拡大すればよい．

```
#declare s = tau * tau / sqrt(3);
object {
   WiredDodecahedron
}
object {
   Icosahedron
   scale <s, s, s>
}
```



問題4の解答　問題3と同様に考えれば，中接球の半径を等しくするためには，正20面体を $\frac{\sqrt{\tau^2+1}}{\sqrt{3}}$ 倍に相似拡大すればよいことがわかる．得られた多面体は，正3角形20枚と正5角形12枚の面をもつ多面体である．この多面体は，**20-12面体** (icosidodecahedron) と呼ばれ，13種類のアルキメデス多面体のうちの1つである．

```
#declare s = sqrt((tau * tau + 1) / 3);
#declare Icosidodecahedron = intersection
{
   object {
      Icosahedron
      scale <s, s, s>
   }
   object {
      Dodecahedron
   }
}
```



問題５の解答

５個の立方体の和を描くと，星形多面体が得られる．

```
union {
   object { SolidCube1 }
   object { SolidCube2 }
   object { SolidCube3 }
   object { SolidCube4 }
   object { SolidCube5 }
}
```



問題６の解答

５個の立方体の共通部分を描くと，**菱形 30 面体** (rhombic triqacontahedron) が得られる．

```
intersection {
   object { SolidCube1 }
   object { SolidCube2 }
   object { SolidCube3 }
   object { SolidCube4 }
   object { SolidCube5 }
}
```



問題7の解答

```
union {
   object { Tetrahedron }
   object { Tetrahedron  scale <-1, -1, -1> }
   object { Tetrahedron2 }
   object { Tetrahedron2  scale <-1, -1, -1> }
   object { Tetrahedron3 }
   object { Tetrahedron3  scale <-1, -1, -1> }
   object { Tetrahedron4 }
   object { Tetrahedron4  scale <-1, -1, -1> }
   object { Tetrahedron5 }
   object { Tetrahedron5  scale <-1, -1, -1> }
}
```



問題8の解答

```
union {
   object { octa1 }
   object { octa2 }
   object { octa3 }
   object { octa4 }
   object { octa5 }
}
```



問題9の解答
問題7の星形多面体と同じものになる.

附録C

5個の立方体の面のデータから，各面の中心を求めることにより，ソリッドな立方体を定義することができる(針金で作った立方体も容易に定義できるので，こちらは省略する).

```
#declare Cube1 = intersection {
   plane { <1, 0, 0>, 1 }
   plane { <-1, 0, 0>, 1 }
   plane { <0, 1, 0>, 1 }
   plane { <0, -1, 0>, 1 }
   plane { <0, 0, 1>, 1 }
   plane { <0, 0, -1>, 1 }
}
#declare Cube2 = intersection {
   plane { <1, tau, inv>, 1 }
   plane { <tau, -inv, -1>, 1 }
```

```
   plane { <-inv, 1, -tau>, 1 }
   plane { <-1, -tau, -inv>, 1 }
   plane { <-tau, inv, 1>, 1 }
   plane { <inv, -1, tau>, 1 }
}
#declare Cube3 = intersection {
   plane { <1, -tau, -inv>, 1 }
   plane { <tau, inv, 1>, 1 }
   plane { <inv, 1, -tau>, 1 }
   plane { <-1, tau, inv>, 1 }
   plane { <-tau, -inv, -1>, 1 }
   plane { <-inv, -1, tau>, 1 }
}
#declare Cube4 = intersection {
   plane { <1, -tau, inv>, 1 }
   plane { <tau, inv, -1>, 1 }
   plane { <inv, 1, tau>, 1 }
   plane { <-1, tau, -inv>, 1 }
   plane { <-tau, -inv, 1>, 1 }
   plane { <-inv, -1, -tau>, 1 }
}
#declare Cube5 = intersection {
   plane { <1, tau, -inv>, 1 }
   plane { <tau, -inv, 1>, 1 }
   plane { <-inv, 1, tau>, 1 }
   plane { <-1, -tau, inv>, 1 }
   plane { <-tau, inv, -1>, 1 }
   plane { <inv, -1, -tau>, 1 }
}
```

附録 D

```
#declare Tetrahedron2 = intersection {
   plane { <-1, -1, 1>, k }
   plane { <tau, 0, inv>, k }
   plane { <-inv, tau, 0 >, k }
   plane { <0, -inv, -tau>, k }
}
#declare Tetrahedron3 = intersection {
   plane { <-1, 1, -1>, k }
```

```
   plane { <tau, 0, -inv>, k }
   plane { <-inv, -tau, 0 >, k }
   plane { <0, inv, tau>, k }
}
#declare Tetrahedron4 = intersection {
   plane { <-1, 1, 1>, k }
   plane { <tau, 0, inv>, k }
   plane { <-inv, -tau, 0 >, k }
   plane { <0, inv, -tau>, k }
}
#declare Tetrahedron5 = intersection {
   plane { <-1, -1, -1>, k }
   plane { <tau, 0, -inv>, k }
   plane { <-inv, tau, 0 >, k }
   plane { <0, -inv, tau>, k }
}
```

附録E

```
#declare octa1 = intersection {
   plane { p1, 1 }
   plane { p2, 1 }
   plane { p3, 1 }
   plane { p4, 1 }
   plane { p5, 1 }
   plane { p6, 1 }
   plane { p7, 1 }
   plane { p8, 1 }
}
#declare octa2 = intersection {
   plane { p2, 1 }
   plane { p7, 1 }
   plane { p9, 1 }
   plane { p12, 1 }
   plane { p14, 1 }
   plane { p15, 1 }
   plane { p17, 1 }
   plane { p20, 1 }
}
#declare octa3 = intersection {
```

```
   plane { p3, 1 }
   plane { p6, 1 }
   plane { p10, 1 }
   plane { p11, 1 }
   plane { p13, 1 }
   plane { p16, 1 }
   plane { p17, 1 }
   plane { p20, 1 }
}
#declare octa4 = intersection {
   plane { p4, 1 }
   plane { p5, 1 }
   plane { p9, 1 }
   plane { p12, 1 }
   plane { p13, 1 }
   plane { p16, 1 }
   plane { p18, 1 }
   plane { p19, 1 }
}
#declare octa5 = intersection {
   plane { p1, 1 }
   plane { p8, 1 }
   plane { p10, 1 }
   plane { p11, 1 }
   plane { p14, 1 }
   plane { p15, 1 }
   plane { p18, 1 }
   plane { p19, 1 }
}
```